Math 5190 − Fixed point theory and the Inverse Function Theorem

Initializations

We define the inverse function contraction function/operator

If x is a list of numbers, then T: x→T[x] implements a function. The y,f,K are parameters
If x=x[y] is an expression/function of y, then T: x → T[x] implements an operator

The first definition of T is for scalar x and y
The second definition is for for vector valued x and y

```
ClearAll[f, y, K, T]
Distance[v1_, v2_] := Sqrt[(v1 - v2).(v1 - v2)]
T[x_, $y_: y, $f_: f, $K_: K] := x + $K ($y - f[x])
T[x_List, $y_: y, $f_: f, $K_: K] := x + $K.($y - f @@ x)
```

An illustration of the NestList function

```
ClearAll[f, y, K];
NestList[f, x, 4]
NestList[T, x, 3] // TableForm
```

```
{x, f[x], f[f[x]], f[f[f[x]]], f[f[f[f[x]]]]}

x
x + K (y - f[x])
x + K (y - f[x]) + K (y - f[x + K (y - f[x])])
x + K (y - f[x]) + K (y - f[x + K (y - f[x])]) + K (y - f[x + K (y - f[x]) + K (y - f[x + K (y - f[x])])])
```

Invert $f : x \to x^2$

```
f = Function[x, x^2];
g = Function[y, Sqrt[y]];
```

contraction FUNCTIONS for some choices of x_0 and y Here x represents a variable number, while y is fixed

Iterations of T produce contraction operators with ever smaller contraction constants
If T is a contraction operator with contraction constant λ, then T^k has contraction constant λ^k
The net result is that T^k for large k contracts the entire domain into a small neighbourhood of the fixed point
Outside the domain there is no convergence

```
x0 = 1;
y = 2.0;
g[y]
NestList[T[#, y, f, (f'[x0])^-1] &, x0, 5]
NestList[T[#, y, f, (f'[x0])^-1] &, x, 2]
NestList[T[#, y, f, (f'[x0])^-1] &, x, 4]
% // Plot[Evaluate[#], {x, -2, 4}, PlotRange → {{-2, 4}, {0, 2}}] &
NestList[T[#, y, f, (f'[x0])^-1] &, x, 4] //
 Plot[Evaluate[#], {x, -2, 0}, PlotLabel → "Explore lower domain of convergence"] &
NestList[T[#, y, f, (f'[x0])^-1] &, x, 4] //
 Plot[Evaluate[#], {x, 2, 4}, PlotLabel → "Explore upper domain of convergence"] &
```

1.41421

{1, 1.5, 1.375, 1.42969, 1.40768, 1.4169}

$$\left\{x,\ x+\frac{1}{2}\left(2.-x^2\right),\ x+\frac{1}{2}\left(2.-x^2\right)+\frac{1}{2}\left(2.-\left(x+\frac{1}{2}\left(2.-x^2\right)\right)^2\right)\right\}$$

$$\left\{x,\ x+\frac{1}{2}\left(2.-x^2\right),\ x+\frac{1}{2}\left(2.-x^2\right)+\frac{1}{2}\left(2.-\left(x+\frac{1}{2}\left(2.-x^2\right)\right)^2\right),\right.$$
$$x+\frac{1}{2}\left(2.-x^2\right)+\frac{1}{2}\left(2.-\left(x+\frac{1}{2}\left(2.-x^2\right)\right)^2\right)+\frac{1}{2}\left(2.-\left(x+\frac{1}{2}\left(2.-x^2\right)+\frac{1}{2}\left(2.-\left(x+\frac{1}{2}\left(2.-x^2\right)\right)^2\right)\right)^2\right),$$
$$x+\frac{1}{2}\left(2.-x^2\right)+\frac{1}{2}\left(2.-\left(x+\frac{1}{2}\left(2.-x^2\right)\right)^2\right)+$$
$$\frac{1}{2}\left(2.-\left(x+\frac{1}{2}\left(2.-x^2\right)+\frac{1}{2}\left(2.-\left(x+\frac{1}{2}\left(2.-x^2\right)\right)^2\right)\right)^2\right)+\frac{1}{2}\left(2.-\left(x+\frac{1}{2}\left(2.-x^2\right)+\right.\right.$$
$$\left.\left.\left.\frac{1}{2}\left(2.-\left(x+\frac{1}{2}\left(2.-x^2\right)\right)^2\right)+\frac{1}{2}\left(2.-\left(x+\frac{1}{2}\left(2.-x^2\right)+\frac{1}{2}\left(2.-\left(x+\frac{1}{2}\left(2.-x^2\right)\right)^2\right)\right)^2\right)\right)^2\right)\right\}$$







Examine the contraction factors

$$\left| \frac{T^{k+1}[x]-T^{k}[x]}{T^{k}[x]-T^{k-1}[x]} \right|$$

Outside the domain of convergence, the contraction ratios can grow above 1

```
T[x, y, f, 1]
NestList[T[#, y, f, (f'[x0])^-1] &, x, 5] //
  Table[Abs[(#[[i + 1]] - #[[i]]) / (#[[i]] - #[[i - 1]])], {i, 2, Length[#] - 1}] & //
 Plot[Evaluate[#], {x, -1, 3.2}, PlotLabel → "Contraction factors", PlotRange → All] &
```

$2. + x - x^2$



Here y is too far away from $y_0 = f[x_0]$. No convergence

```
x0 = 1;
y = 6.0;
g[y]
NestList[T[#, y, f, (f'[x0])^-1] &, x0, 5]
NestList[T[#, y, f, (f'[x0])^-1] &, x, 10] //
 Plot[Evaluate[#], {x, -2, 4}, PlotRange → {{-2, 4}, {0, 2}}] &
```

2.44949

{1, 3.5, 0.375, 3.30469, 0.844208, 3.48786}



Get convergence by choosing x_0 so that $y_0 = f[x_0]$ is suffciently close to y

```
x0 = 4;
y = 6.0;
g[y]
NestList[T[#, y, f, (f'[x0])^-1] &, x0, 5]
NestList[T[#, y, f, (f'[x0])^-1] &, x, 5] // Plot[Evaluate[#], {x, -3, 11}, PlotRange → All] &
NestList[T[#, y, f, (f'[x0])^-1] &, x, 8] //
  Table[Abs[(#[[i + 1]] - #[[i]])/(#[[i]] - #[[i - 1]])], {i, 2, Length[#] - 1}] & //
 Plot[Evaluate[#], {x, -2, 10}, PlotLabel → "Contraction factors", PlotRange → All] &
```

```
2.44949
```

```
{4, 2.75, 2.55469, 2.48888, 2.46457, 2.45531}
```





contraction OPERATORS for some choices of x_0 Here x represents a function/expression of a variable y

First, we see a comparison of the series expansions of the approximate inverses and the true inverse
Second, we plot the approximate inverses (these are now functions of y)
and see how they converge to a well defined function with higher iterations
Then, we take all of the above functions and subtract them from the true inverse g[y].
The results should converge to the 0 function

```
x0 = 1;
ClearAll[y]
Series[g[y], {y, f[x0], 6}]
NestList[T[#, y, f, (f'[x0])^-1] &, x0, 3] // Series[#, {y, f[x0], 6}] &
NestList[T[#, y, f, (f'[x0])^-1] &, x0, 6] //
 Plot[Evaluate[#], {y, -1, 3}, PlotLabel → "Show the convergence of the functions"] &
NestList[T[#, y, f, (f'[x0])^-1] &, x0, 6] // Plot[Evaluate[# - g[y]], {y, 0, 3},
   PlotLabel → "Show the convergence of the functions relative to g[y]"] &
```

$$1+\frac{y-1}{2}-\frac{1}{8}(y-1)^2+\frac{1}{16}(y-1)^3-\frac{5}{128}(y-1)^4+\frac{7}{256}(y-1)^5-\frac{21(y-1)^6}{1024}+O[y-1]^7$$

$$\left\{1,\ 1+\frac{y-1}{2}+O[y-1]^7,\ 1+\frac{y-1}{2}-\frac{1}{8}(y-1)^2+O[y-1]^7,\right.$$
$$\left.1+\frac{y-1}{2}-\frac{1}{8}(y-1)^2+\frac{1}{16}(y-1)^3-\frac{1}{128}(y-1)^4+O[y-1]^7\right\}$$





Let’s try a different x_0
We throw away the first approximation because it’s so lousy (this is what the Take function does)
Again, we see that convergence only happens in a limited domain, and that the rate of convergence
improves if we restrict the domains

```
x0 = 2;
ClearAll[y]
Series[g[y], {y, f[x0], 6}]
NestList[T[#, y, f, (f'[x0])^-1] &, x0, 3] // Series[#, {y, f[x0], 6}] &
NestList[T[#, y, f, (f'[x0])^-1] &, x0, 6] // Take[#, -5] & //
 Plot[Evaluate[# - g[y]], {y, 0, 10}, PlotLabel → "Convergence rel to g[y]"] &
```

$$2+\frac{y-4}{4}-\frac{1}{64}(y-4)^2+\frac{1}{512}(y-4)^3-\frac{5(y-4)^4}{16384}+\frac{7(y-4)^5}{131072}-\frac{21(y-4)^6}{2097152}+O[y-4]^7$$

$$\left\{2,\ 2+\frac{y-4}{4}+O[y-4]^7,\ 2+\frac{y-4}{4}-\frac{1}{64}(y-4)^2+O[y-4]^7,\right.$$
$$\left.2+\frac{y-4}{4}-\frac{1}{64}(y-4)^2+\frac{1}{512}(y-4)^3-\frac{(y-4)^4}{16384}+O[y-4]^7\right\}$$



- **Invert $f : x \to \mathrm{Sin}[x]$**

```
f = Function[x, Sin[x]];
g = Function[y, ArcSin[y]];
```

- **contraction FUNCTIONS for some choices of $x_0$and y**
  Here we approximate ArcSin using the Sin function

```
Plot some initial approximation formulas and then plot the iterates of the contrac-
tion functions
Here y is a fixed constant, so we are considering functions that map some x-domain
close to the unknown
fixedpoint ArcSin[y], y fixed
Then, we plot the contraction factors
```

```
x0 = 0;
y = 0.4;
g[y]
NestList[T[#, y, f, (f'[x0])^-1] &, x0, 10]
NestList[T[#, y, f, (f'[x0])^-1] &, x, 3] // TableForm
NestList[T[#, y, f, (f'[x0])^-1] &, x, 8] // Plot[Evaluate[#], {x, -3.2, 3.2}] &
NestList[T[#, y, f, (f'[x0])^-1] &, x, 5] //
  Table[Abs[(#[[i + 1]] - #[[i]])/(#[[i]] - #[[i - 1]])], {i, 2, Length[#] - 1}] & //
 Plot[Evaluate[#], {x, -3.2, 3.2}, PlotLabel → "Contraction factors"] &
```

```
0.411517
```

```
{0, 0.4, 0.410582, 0.411439, 0.41151,
 0.411516, 0.411517, 0.411517, 0.411517, 0.411517, 0.411517}
```

```
x
0.4 + x - Sin[x]
0.8 + x - Sin[x] - Sin[0.4 + x - Sin[x]]
1.2 + x - Sin[x] - Sin[0.4 + x - Sin[x]] - Sin[0.8 + x - Sin[x] - Sin[0.4 + x - Sin[x]]]
```





Here y is further away from $y_0 = f[x_0]$. Convergence is slower, the contraction constant closer to 1.
Here we see a curious phenomenon. For x sufficiently far away from the given x_0 we have convergence to another fixed point
(Recall that ArcSin is a multi-valued function)
This does not violate the Fixed Point Theorem, because on the larger domain, the contraction constant is not bounded by 1.

```
x0 = 0;
y = 0.9;
g[y]
NestList[T[#, y, f, (f'[x0])^-1] &, x, 8] // Plot[Evaluate[#], {x, -3.2, 6.4}] &
NestList[T[#, y, f, (f'[x0])^-1] &, x, 5] //
  Table[Abs[(#[[i + 1]] - #[[i]]) / (#[[i]] - #[[i - 1]])], {i, 2, Length[#] - 1}] & //
 Plot[Evaluate[#], {x, -3.2, 6.4}, PlotLabel -> "Contraction factors"] &
```

1.11977





Here y is too far away from $y_0 = f[x_0]$. No convergence. The contraction factors are note bounded by 1

```
x0 = 0;
y = 2.0;
g[y]
NestList[T[#, y, f, (f'[x0])^-1] &, x, 8] // Plot[Evaluate[#], {x, -3.2, 6.4}] &
NestList[T[#, y, f, (f'[x0])^-1] &, x, 5] //
  Table[Abs[(#[[i + 1]] - #[[i]]) / (#[[i]] - #[[i - 1]])], {i, 2, Length[#] - 1}] & //
 Plot[Evaluate[#], {x, -3.2, 6.4}, PlotLabel → "Contraction factors"] &
```

1.5708 − 1.31696 ⅈ





- **contraction OPERATOR for some choices of x_0**
  Here x is an expression/function of a variable y
  Iterates of the operator T give us approximations to the inverse functions
  First, we compare the power series of the approximations to the actual power series,
  Second, we plot the approximate inverse
  Third, we plot the difference between the approximate inverses and the true inverse

```
NestList[T[#, y, f, (f'[x0])^-1] &, x0, 4]
Plot[Evaluate[Join[{Arcsin[y]}, %]], {y, -2, 2}]
```

```
{0, y, 2 y - Sin[y], 3 y - Sin[y] - Sin[2 y - Sin[y]],
 4 y - Sin[y] - Sin[2 y - Sin[y]] - Sin[3 y - Sin[y] - Sin[2 y - Sin[y]]]}
```



```
x0 = 0;
ClearAll[y]
Series[g[y], {y, f[x0], 6}]
NestList[T[#, y, f, (f'[x0])^-1] &, x0, 3] // Series[#, {y, f[x0], 6}] &
NestList[T[#, y, f, (f'[x0])^-1] &, x0, 6] //
 Plot[Evaluate[#], {y, -2, 2}, PlotLabel → "Show the convergence of the functions"] &
NestList[T[#, y, f, (f'[x0])^-1] &, x0, 6] // Plot[Evaluate[# - g[y]], {y, -1, 1},
   PlotLabel → "Show the convergence of the functions relative to ArcSin[y]"] &
```

$y + \frac{y^3}{6} + \frac{3 y^5}{40} + O[y]^7$

$\left\{0,\ y + O[y]^7,\ y + \frac{y^3}{6} - \frac{y^5}{120} + O[y]^7,\ y + \frac{y^3}{6} + \frac{3 y^5}{40} + O[y]^7\right\}$





- **Multivariable IFT**
  Invert $f : \{x, y\} \to \{x^3 - 3 x y^2, -3 x^2 y + y^3\}$

Here f is a vector valued function of 2 variables
Jf is the Jacobian
g is the uknown inverse (*Mathematica* can approximate it numerically, but a closed form expression is quite complicated)

```
f = Function[{x1, x2}, {x1^3 - 3 x1 x2^2, -3 x1^2 x2 + x2^3}];
Jf = Function[{x1, x2}, {{3 x1^2 - 3 x2^2, -6 x1 x2}, {-6 x1 x2, -3 x1^2 + 3 x2^2}}];
f[x1, x2]
Jf[x1, x2] // MatrixForm
ClearAll[g]
```

$\{x1^3 - 3\,x1\,x2^2, -3\,x1^2\,x2 + x2^3\}$

$$\begin{pmatrix} 3\,x1^2 - 3\,x2^2 & -6\,x1\,x2 \\ -6\,x1\,x2 & -3\,x1^2 + 3\,x2^2 \end{pmatrix}$$

- **contraction FUNCTIONS for some choices of y**

First we ask *Mathematica* to find the inverse for this one particular y numerically

```
x0 = {1, 1};
f @@ x0
y = {-2.5, -3.0};
Solve[f @@ {x1, x2} == y, {x1, x2}] // TableForm
```

{-2, -2}

x1 → -1.50808	x2 → 0.45335
x1 → -0.573327 - 0.993031 i	x2 → -0.539681 - 0.934756 i
x1 → -0.573327 + 0.993031 i	x2 → -0.539681 + 0.934756 i
x1 → -0.180714 + 0.313007 i	x2 → 0.766356 - 1.32737 i
x1 → -0.180714 - 0.313007 i	x2 → 0.766356 + 1.32737 i
x1 → 0.361429	x2 → -1.53271
x1 → 0.754041 - 1.30604 i	x2 → -0.226675 + 0.392612 i
x1 → 0.754041 + 1.30604 i	x2 → -0.226675 - 0.392612 i
x1 → 1.14665	x2 → 1.07936

We calculate the inverse of the Jacobian matrix at x_0 (this is the matrix $K)
We then use this matrix to define a contraction FUNCTION
The contraction function iteratively applied to a particular value of $x_0 = \{x1, x2\}$
We obtain an approximation of one of the above numerical solutions

```
x0 = {1, 1};
$K = Inverse[Jf @@ x0];
$K // MatrixForm
NestList[T[#, y, f, $K] &, x0, 5]
```

$$\begin{pmatrix} 0 & -\frac{1}{6} \\ -\frac{1}{6} & 0 \end{pmatrix}$$

{{1, 1}, {1.16667, 1.08333}, {1.1413, 1.08005},
 {1.14786, 1.07882}, {1.14641, 1.07958}, {1.1467, 1.07929}}

Plot the contraction functions for various iterates of T
Since we are dealing with 2 functions of 2 variables (unplottable) we plot the distance from the approximate fixed point
over a rectangular region
The animation shows how higher iterates of the contraction operator T "drive" the rectangular region in question
towards a single fixed point value

```
x0 = {1, 1};
$K = Inverse[Jf @@ x0];
Nest[T[#, y, f, $K] &, {x1, x2}, 2]
Nest[T[#, y, f, $K] &, x0, 6]
ListAnimate[
 Map[Plot3D[Evaluate[Distance[%, #]], {x1, 0, 2}, {x2, 0, 2}, DisplayFunction → Identity,
    PlotRange → {0, 2}] &, NestList[T[#, y, f, $K] &, {x1, x2}, 6]]]
```

$$\left\{x1+\frac{1}{6}\left(3.-3\,x1^2\,x2+x2^3\right)+\frac{1}{6}\left(3.+\left(x2+\frac{1}{6}\left(2.5+x1^3-3\,x1\,x2^2\right)\right)^3-3\left(x2+\frac{1}{6}\left(2.5+x1^3-3\,x1\,x2^2\right)\right)\left(x1+\frac{1}{6}\left(3.-3\,x1^2\,x2+x2^3\right)\right)^2\right),\right.$$
$$\left.x2+\frac{1}{6}\left(2.5+x1^3-3\,x1\,x2^2\right)+\frac{1}{6}\left(2.5-3\left(x2+\frac{1}{6}\left(2.5+x1^3-3\,x1\,x2^2\right)\right)^2\left(x1+\frac{1}{6}\left(3.-3\,x1^2\,x2+x2^3\right)\right)+\left(x1+\frac{1}{6}\left(3.-3\,x1^2\,x2+x2^3\right)\right)^3\right)\right\}$$

{1.14665, 1.07938}



Here y is too far away from $y_0 = f[x_0]$.
The resulting function T is not a contraction operator and there is no convergence

```
x0 = {1, 1};
f @@ x0
y = {-8.0, -5.0};
Solve[f @@ {x1, x2} == y, {x1, x2}] // TableForm
$K = Inverse[Jf @@ x0];
NestList[T[#, y, f, $K] &, x0, 10]
```

```
{-2, -2}
```

x1 → -2.07647	x2 → 0.39117
x1 → -0.688499 - 1.19252 i	x2 → -0.801346 - 1.38797 i
x1 → -0.688499 + 1.19252 i	x2 → -0.801346 + 1.38797 i
x1 → -0.349736 + 0.605761 i	x2 → 0.996931 - 1.72673 i
x1 → -0.349736 - 0.605761 i	x2 → 0.996931 + 1.72673 i
x1 → 0.699473	x2 → -1.99386
x1 → 1.03824 - 1.79828 i	x2 → -0.195585 + 0.338763 i
x1 → 1.03824 + 1.79828 i	x2 → -0.195585 - 0.338763 i
x1 → 1.377	x2 → 1.60269

```
{{1, 1}, {1.5, 2.}, {1.41667, 0.895833}, {1.47088, 2.13458},
 {1.61616, 0.64731}, {1.64932, 2.34561}, {1.44321, -0.110487},
 {2.39138, 1.71503}, {-0.838412, 1.8107}, {0.347958, 4.42023}, {15.3078, 2.36131}}
```

- **contraction OPERATORS for some choices of x_0**
  Here x is an expression/function of the variable y

First, we calculate the approximate inverses. We get some rather nasty, but polynomial functions of y
Then, compose them with the given f[x1,x2] and display the distance to the identity function

```
x0 = {1, 1};
f @@ x0
$K = Inverse[Jf @@ x0];
NestList[Together[T[#, {y1, y2}, f, $K]] &, x0, 2]
```

```
{-2, -2}
```

$$\left\{\{1, 1\}, \left\{\frac{4 - y2}{6}, \frac{4 - y1}{6}\right\}, \left\{\frac{736 + 12\, y1^2 - y1^3 - 336\, y2 - 24\, y1\, y2 - 12\, y2^2 + 3\, y1\, y2^2}{1296}, \frac{736 - 336\, y1 - 12\, y1^2 - 24\, y1\, y2 + 3\, y1^2\, y2 + 12\, y2^2 - y2^3}{1296}\right\}\right\}$$

If the above functions $x = g_k[y]$ are approximate inverses, then the compositions them with $f[g_k[y]]$ should approximately equal y
We compose the approximate inverses calculated above and plot the distance of the result from the identity function
We throw away the first approximation using the Take function (it’s too inaccurate to be compared with the subsequent iterates)

```
NestList[Together[T[#, {y1, y2}, f, $K]] &, x0, 1]
Map[Distance[(f @@ #), {y1, y2}] &,
 Take[NestList[Together[T[#, {y1, y2}, f, $K]] &, x0, 1], -1]]
Map[Plot3D[Evaluate[#], {y1, -2.5, -2}, {y2, -2.5, -2}, PlotRange → {0, 0.01}] &, Map[
  Distance[(f @@ #), {y1, y2}] &, Take[NestList[Together[T[#, {y1, y2}, f, $K]] &, x0, 4], -3]]]
```

$$\left\{\{1, 1\}, \left\{\frac{4 - y2}{6}, \frac{4 - y1}{6}\right\}\right\}$$

$$\left\{\sqrt{\left(-y1-\frac{1}{72}(4-y1)^2(4-y2)+\frac{1}{216}(4-y2)^3\right)^2+\left(\frac{1}{216}(4-y1)^3-\frac{1}{72}(4-y1)(4-y2)^2-y2\right)^2}\right\}$$